Population with several species - Systems of ODE

```
> restart;
> with(DEtools): with(plots):
```

A linear model for 2 species

```
> twolin := { diff(p(t),t) = b1 + c1*p(t) + d1*q(t),
              diff(q(t),t) = b2 + c2*p(t) + d2*q(t) };
```

$$twolin := \left\{ \frac{\mathrm{d}}{\mathrm{d}t}\, p(t) = b1 + c1\, p(t) + d1\, q(t),\ \frac{\mathrm{d}}{\mathrm{d}t}\, q(t) = b2 + c2\, p(t) + d2\, q(t) \right\} \tag{1.1}$$

Example: arms race - steady state scenario

```
> linex1 := subs( {b1=3/2, c1=-1,  d1=1/2,
                   b2=0,   c2=1/2, d2=-1},
                  twolin);
```

$$linex1 := \left\{ \frac{\mathrm{d}}{\mathrm{d}t}\, p(t) = \frac{3}{2} - p(t) + \frac{1}{2}\, q(t),\ \frac{\mathrm{d}}{\mathrm{d}t}\, q(t) = \frac{1}{2}\, p(t) - q(t) \right\} \tag{1.1.1}$$

```
> linsol1 := dsolve( { op(linex1), p(0)=1/2, q(0)=3/2 }, {p
(t), q(t)});
```

$$linsol1 := \left\{ p(t) = -\frac{1}{2}\, \mathrm{e}^{-\frac{1}{2}t} - \mathrm{e}^{-\frac{3}{2}t} + 2,\ q(t) = -\frac{1}{2}\, \mathrm{e}^{-\frac{1}{2}t} + \mathrm{e}^{-\frac{3}{2}t} + 1 \right\} \tag{1.1.2}$$

```
> cntry1 := unapply( subs(linsol1, p(t)), t):
cntry2 := unapply( subs(linsol1, q(t)), t):
plot([cntry1,cntry2],0..7,arms=0..2,thickness=3);
```



Example: arms race - unlimited growth

```
> linex2 := subs( {b1=1/2,  c1=-3/4,  d1=1,
                    b2=-5/4, c2=1,     d2=-3/4},
                   twolin);
```

$$linex2 := \left\{ \frac{d}{dt}\, p(t) = \frac{1}{2} - \frac{3}{4}\, p(t) + q(t),\ \frac{d}{dt}\, q(t) = -\frac{5}{4} + p(t) - \frac{3}{4}\, q(t) \right\} \quad \textbf{(1.2.1)}$$

```
> linsol2 := dsolve( { op(linex2), p(0)=1, q(0)=5/2 }, {p(t),
  q(t)}):
```

```
> cntry1 := unapply( subs(linsol2, p(t)), t):
  cntry2 := unapply( subs(linsol2, q(t)), t):
  plot([cntry1,cntry2],0..10,p=0..5,thickness=3);
```



Example: the peaceful neighbour

Parameter c2 is **negative**, so country 2 will destroy weapons, if its neighbour's armament grows.

```
> linex3 := subs( {b1=0,  c1=-3/4,  d1=1,
                   b2=5/2, c2=-1,     d2=-3/4},
                  twolin);
```

$$linex3 := \left\{ \frac{d}{dt}\, p(t) = -\frac{3}{4}\, p(t) + q(t),\ \frac{d}{dt}\, q(t) = \frac{5}{2} - p(t) - \frac{3}{4}\, q(t) \right\} \quad \textbf{(1.3.1)}$$

```
> linsol3 := dsolve( { op(linex3), p(0)=1/2, q(0)=3 }, {p(t),
  q(t)}):
```

```
> cntry1 := unapply( subs(linsol3, p(t)), t):
  cntry2 := unapply( subs(linsol3, q(t)), t):
  plot([cntry1,cntry2],0..10,p=0..5,thickness=3);
```



A non-linear model for 2 species

Here, we start from the logistic equation, and add a term for the other species:

```
> twolv := { diff(p(t),t) = ( a1 + b1*p(t) + c1*q(t) ) * p(t),
             diff(q(t),t) = ( a2 + b2*p(t) + c2*q(t) ) * q(t) }
  ;
```

$$twolv := \left\{ \frac{d}{dt}\, p(t) = (a1 + b1\, p(t) + c1\, q(t))\, p(t),\ \frac{d}{dt}\, q(t) = (a2 + b2\, p(t) + c2\, q(t))\, q(t) \right\} \qquad \textbf{(2.1)}$$

Example: Competition - Steady State

Competition between two species (sharing common ressources, for example)

```
> compex1 := subs( { a1=5/2+sqrt(3)/24, b1=-5/8,        c1=-
  sqrt(3)/24,
                     a2=7/8+sqrt(27)/2, b2=-sqrt(27)/8, c2=
  -7/8},
                  twolv);
```

(2.1.1)

$$compex1 := \left\{ \frac{d}{dt} p(t) = \left(\frac{5}{2} + \frac{1}{24}\sqrt{3} - \frac{5}{8} p(t) - \frac{1}{24}\sqrt{3}\, q(t) \right) p(t), \frac{d}{dt} q(t) = \left(\frac{7}{8} + \frac{3}{2}\sqrt{3} - \frac{3}{8}\sqrt{3}\, p(t) - \frac{7}{8} q(t) \right) q(t) \right\} \quad \textbf{(2.1.1)}$$

```
> spred := DEplot( compex1, [p,q], t=0..10, [[p(0)=1/4, q(0)=
3]],
                   stepsize=0.2, scene=[t,p], linecolor=red):
  spgre := DEplot( compex1, [p,q], t=0..10, [[p(0)=1/4, q(0)=
3]],
                   stepsize=0.2, scene=[t,q], linecolor=
green):
  display([spred,spgre]);
```



Example: Competition - Extinction

In this example, one species dies out:

```
> compex2 := subs( {a1=71/8, b1=-23/12, c1=-25/12,
                    a2=73/8, b2=-25/12, c2=-23/12},
                   twolv);
```

(2.2.1)

$$compex2 := \left\{ \frac{d}{dt} p(t) = \left(\frac{71}{8} - \frac{23}{12} p(t) - \frac{25}{12} q(t) \right) p(t), \frac{d}{dt} q(t) = \left(\frac{73}{8} - \frac{25}{12} p(t) - \frac{23}{12} q(t) \right) q(t) \right\} \quad \textbf{(2.2.1)}$$

```
> spred := DEplot( compex2, [p,q], t=0..10, [[p(0)=1/4, q(0)=1/2]],
                    stepsize=0.2, scene=[t,p], linecolor=red):
  spgre := DEplot( compex2, [p,q], t=0..10, [[p(0)=1/4, q(0)=1/4]],
                    stepsize=0.2, scene=[t,q], linecolor=green):
  display([spred,spgre]);
```



A non-linear model for 2 species (predator-prey): Volterra-Lotka

```
> twospec := { diff(p(t),t) = ( b1 + c1*q(t)) * p(t),
               diff(q(t),t) = ( b2 + c2*p(t)) * q(t) };
```

(3.1)

$$twospec := \left\{ \frac{d}{dt}\, p(t) = (b1 + c1\, q(t))\, p(t),\ \frac{d}{dt}\, q(t) = (b2 + c2\, p(t))\, q(t) \right\} \quad (3.1)$$

```
> tsex1 := subs( {b1=-1/2, b2=1/5,
                 c1=1/200, c2=-1/50},
                 twospec);
```

$$tsex1 := \left\{ \frac{d}{dt}\, p(t) = \left(-\frac{1}{2} + \frac{1}{200}\, q(t)\right) p(t),\ \frac{d}{dt}\, q(t) = \left(\frac{1}{5} - \frac{1}{50}\, p(t)\right) q(t) \right\} \quad (3.2)$$

The non-linear equation is too difficult to be solved by Maple.
We use Maple's numerical method, instead:

```
> pred := DEplot( tsex1, [p,q], t=0..100, [[p(0)=6, q(0)=50]],
                  stepsize=0.2, scene=[t,p], linecolor=red):
> prey := DEplot( tsex1, [p,q], t=0..100, [[p(0)=6, q(0)=50]],
                  stepsize=0.2, scene=[t,q], linecolor=blue):
> display([pred,prey]);
```